
1 Introduction 
This document describes the Smart city RDF benchmark for the performance evaluation of RDF stores for 

Smart city services. 

2 SPARQL Queries 
In the following sections are reported the queries to be used for the benchmark. Since each RDF store has a 

specific extension for full text search and geo-spatial search a line starting with #VOS should be used only 

when using Virtuoso and #OIM only for GraphDB/OWLIM. The queries may depend on some parameters as:  

 %LAT %LNG for a geo-spatial position 

 %LINE for a bus line 

 %BUS_STOP for  a bus stop 

 %DIST for a distance in km 

 %SENSOR for a sensor identifier 

 %WORD for a word to be used for text search 

Values to be used for these parameters are reported in the parameters data section in the following 

2.1 Find-address 
given the latitude and longitude position %LAT %LNG retrieves the nearest address within 100m. 

#ID:find-address 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX vcard:<http://www.w3.org/2006/vcard/ns#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT DISTINCT ?via ?numero ?comune WHERE { 

  ?entry rdf:type km4c:Entry. 

  ?nc km4c:hasExternalAccess ?entry. 

  ?nc km4c:extendNumber ?numero. 

  ?nc km4c:belongToRoad ?road. 

  ?road km4c:extendName ?via. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

  ?road km4c:inMunicipalityOf ?com. 

  ?com foaf:name ?comune. 



#VOS ?entry geo:geometry ?geo. 

#VOS filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=0.1) 

#VOS BIND( bif:st_distance(?geo, bif:st_point(%LNG, %LAT)) AS ?dist) 

#OIM ?entry omgeo:nearby(%LAT %LNG "0.1km"). 

#OIM BIND( omgeo:distance(?elat, ?elong, %LAT, %LNG) AS ?dist) 

} ORDER BY ?dist LIMIT 1 

2.2 Municipalities-florence 
retrieves the list of municipalities within the province of Florence. 

#ID:municipalities-florence 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

SELECT DISTINCT ?mun ?nomeComune WHERE { 

 ?mun rdf:type km4c:Municipality. 

 ?mun km4c:isPartOfProvince ?prov. 

 ?prov foaf:name "FIRENZE". 

 ?mun foaf:name ?nomeComune. 

} 

ORDER BY ?nomeComune 

2.3 Bus-lines 
retrieve the list of bus lines. 

#ID:bus-lines 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

select distinct ?n where { 

 ?x rdf:type km4c:PublicTransportLine; 

 dcterms:identifier ?n. 

 bind(xsd:integer(?n) as ?d) 

} order by ?d 

2.4 Bus-stops-of-line 
given the bus line %LINE retrieves the bus stops list of the line. 

#ID:bus-stops-of-line 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT DISTINCT ?bs ?bslat ?bslong ?nomeFermata ?x WHERE { 

 ?tpll rdf:type km4c:PublicTransportLine. 

 ?tpll dcterms:identifier "%LINE". 

 ?tpll km4c:hasRoute ?route. 

 ?route km4c:hasSection ?rs. 

 ?rs km4c:startsAtStop ?bs. 

 ?bs foaf:name ?nomeFermata. 

 ?bs geo:lat ?bslat. 

 ?bs geo:long ?bslong. 

} ORDER BY ?nomeFermata 



 

Note: the ?x unused variable is needed by virtuoso to overcome a bug in the DISTINCT clause 

2.5 Lines-of-bus-stop 
given a bus stop name %BUS_STOP retrieves the lines passing from the same bus stop. 

#ID:lines-of-stop 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX km4cr:<http://www.disit.org/km4city/resource/> 

PREFIX schema:<http://schema.org/#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX vcard:<http://www.w3.org/2006/vcard/ns#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT DISTINCT ?id WHERE{ 

 ?tpll rdf:type km4c:PublicTransportLine. 

 ?tpll dcterms:identifier ?id. 

 ?tpll km4c:hasRoute ?route. 

 ?route km4c:hasSection ?rs. 

 ?rs km4c:endsAtStop ?bs1. 

 ?rs km4c:startsAtStop ?bs2. 

 {?bs1 foaf:name "%BUS_STOP".} 

 UNION 

 {?bs2 foaf:name "%BUS_STOP".} 

} ORDER BY ?id 

2.6 Bus-stop-latlng 
given a position %LAT, %LNG and a radius %DIST finds the bus stops that are within the radius. 

#ID:bus-stop-latlng-%DISTkm 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

SELECT DISTINCT ?bs (STR(?nome) AS ?bsName) ?bslat ?bslong ?x WHERE { 

 ?bs rdf:type km4c:BusStop. 

 ?bs foaf:name ?nome. 

 ?bs geo:lat ?bslat. 

 ?bs geo:long ?bslong. 

#VOS ?bs geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM ?bs omgeo:nearby(%LAT %LNG "%DISTkm"). 

} 

 

Note: the ?x unused variable is needed by virtuoso to overcome a bug in the DISTINCT clause 

2.7 Bus-stop-florence 
retrieves all the bus stops in the municipality of Florence. 

#ID:bus-stop-florence 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 



PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

SELECT DISTINCT ?bs ?nomeFermata ?bslat ?bslong ?x WHERE { 

 ?bs rdf:type km4c:BusStop. 

 ?bs foaf:name ?nomeFermata. 

 ?bs geo:lat ?bslat. 

 ?bs geo:long ?bslong. 

 ?bs km4c:isInMunicipality ?com. 

 ?com foaf:name "FIRENZE". 

} 

 

Note: the ?x unused variable is needed by virtuoso to overcome a bug in the DISTINCT clause 

2.8 Bus-stop-forecast 
given a bus stop %BUS_STOP finds the next forecasts for the lines crossing at the bus stop. 

#ID:bus-stop-forecast 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX time:<http://www.w3.org/2006/time#> 

SELECT DISTINCT ?avmr ?line ?state ?arrivalTime ?idRide WHERE { 

 { 

  SELECT ?ride (MAX(?avmr) AS ?avmrLast) WHERE{ 

   ?bs rdf:type km4c:BusStop. 

   ?bs foaf:name "%BUS_STOP". 

   ?bs km4c:hasForecast ?bsf. 

   ?avmr km4c:includeForecast ?bsf. 

   ?ride km4c:hasAVMRecord ?avmr. 

  }GROUP BY ?ride ORDER BY DESC (?avmrLast) LIMIT 15 

 } 

 ?bs rdf:type km4c:BusStop. 

 ?bs foaf:name "%BUS_STOP". 

 ?bs km4c:hasForecast ?forecast. 

 ?avmrLast km4c:includeForecast ?forecast. 

 ?forecast km4c:expectedTime ?arrivalTime. 

 ?avmrLast km4c:concernLine ?line. 

 ?avmrLast km4c:rideState ?state. 

 ?ride km4c:hasAVMRecord ?avmrLast. 

 ?ride dcterms:identifier ?idRide. 

 FILTER(?arrivalTime>=xsd:dateTime("2015-05-06T09:00:00+02:00")) 

} ORDER BY ?arrivalTime 

2.9 AVM-distribution 
retrieves for each day the count of AVM records received. 

#ID:avm-distribution 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT ?yy ?mm ?dd (count(*) as ?c) where { 

 ?x a km4c:AVMRecord. 



 ?x dcterms:created ?d. 

 BIND( year(?d) as ?yy) 

 BIND( month(?d) as ?mm) 

 BIND( day(?d) as ?dd) 

} GROUP BY ?yy ?mm ?dd ORDER BY ?yy ?mm ?dd 

2.10 Service-florence 
retrieves all the services in the municipality of Florence. 

#ID:service-florence 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

SELECT DISTINCT ?ser ?serAddress ?serNumber ?elat ?elong ?sName ?sType ?email 

?note ?labelIta ?multimedia ?description ?identifier ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS  OPTION (inference "urn:ontology"). 

 OPTIONAL{?ser schema:name ?sName. } 

 ?ser schema:streetAddress ?serAddress. 

 OPTIONAL {?ser km4c:houseNumber ?serNumber}. 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it")} 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia } 

 OPTIONAL { ?ser dcterms:identifier ?identifier } 

 OPTIONAL {?ser skos:note ?note } 

 OPTIONAL {?ser schema:email ?email } 

 {  

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

  ?nc km4c:hasExternalAccess ?entry. 

  ?nc km4c:belongToRoad ?road. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } UNION { 

  ?ser km4c:isInRoad ?road . 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?labelIta. FILTER(LANG(?labelIta)="it") 

} 

 



Note: the ?x unused variable is needed by virtuoso to overcome a bug in the DISTINCT clause 

2.11 Service-Acc-Clt-Trs-W&F-florence 
retrieves all the services in the Accomodation, Clutural Activity, TourismService and Wine&Food classes 

within the municipality of Florence. 

#ID:service-acc-clt-trs-w&f-florence 
PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX km4cr:<http://www.disit.org/km4city/resource#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

SELECT DISTINCT ?ser ?serAddress ?serNumber ?elat ?elong ?sName ?sType ?email 

?note ?labelIta ?multimedia ?description ?identifier ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS  OPTION (inference "urn:ontology"). 

 OPTIONAL{?ser schema:name ?sName. } 

 ?ser schema:streetAddress ?serAddress. 

 OPTIONAL {?ser km4c:houseNumber ?serNumber}. 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it")} 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia } 

 OPTIONAL {?ser dcterms:identifier ?identifier } 

 OPTIONAL {?ser skos:note ?note } 

 OPTIONAL {?ser schema:email ?email } 

 {  

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

  ?nc km4c:hasExternalAccess ?entry. 

  ?nc km4c:belongToRoad ?road. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } UNION { 

  ?ser km4c:isInRoad ?road . 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } 

 { ?ser a km4c:Accommodation 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:CulturalActivity 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:TourismService 

#VOS OPTION (inference "urn:ontology"). 

 } 



UNION { ?ser a km4c:WineAndFood 

#VOS OPTION (inference "urn:ontology"). 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?labelIta. FILTER(LANG(?labelIta)="it") 

} 

2.12 Service-Htl-B&B-florence 
retrieves all the services in the Hotel and Bed&Breakfast classes within the municipality of Florence. 

#ID:service-htl-b&b-florence 
PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

SELECT DISTINCT ?ser ?serAddress ?serNumber ?elat ?elong ?sName ?sType ?email 

?note ?labelIta ?multimedia ?description ?identifier ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS OPTION (inference "urn:ontology"). 

 OPTIONAL{?ser schema:name ?sName. } 

 ?ser schema:streetAddress ?serAddress. 

 OPTIONAL {?ser km4c:houseNumber ?serNumber}. 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it")} 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia } 

 OPTIONAL {?ser dcterms:identifier ?identifier } 

 OPTIONAL {?ser skos:note ?note } 

 OPTIONAL {?ser schema:email ?email } 

 {  

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

  ?nc km4c:hasExternalAccess ?entry. 

  ?nc km4c:belongToRoad ?road. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } UNION { 

  ?ser km4c:isInRoad ?road . 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

  ?road km4c:inMunicipalityOf ?mun. 

  ?mun foaf:name "FIRENZE". 

 } 

 { ?ser a km4c:Hotel 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:Bed_and_breakfast 



#VOS OPTION (inference "urn:ontology"). 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?labelIta. FILTER(LANG(?labelIta)="it") 

} 

2.13 Service-latlng 
retrieves the services within a radius from a latitude, longitude position. 

#ID:service-latlng-%DISTkm 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX schema:<http://schema.org/> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

SELECT DISTINCT ?ser ?serAddress ?elat ?elong ?sType ?sTypeIta ?sName ?email 

?note ?multimedia ?description ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS  OPTION (inference "urn:ontology"). 

 OPTIONAL {?ser <http://schema.org/name> ?sName} 

 ?ser <http://schema.org/streetAddress> ?serAddress. 

 OPTIONAL {?ser skos:note ?note} 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it") } 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia} 

 OPTIONAL {?ser schema:email ?email } 

 { 

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

#VOS  ?entry geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?entry omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } UNION { 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

#VOS  ?ser geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?ser omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?sTypeIta. FILTER(LANG(?sTypeIta)="it") 

} 



2.14 Service-Acc-Clt-Trs-W&F-latlng 
retrieves all the services in the Accomodation, Clutural Activity, TourismService and Wine&Food classes 

within a radius %DIST from a position %LAT %LNG. 

#ID:service-acc-clt-trs-w&f-latlng-%DISTkm 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX schema:<http://schema.org/> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

SELECT DISTINCT ?ser ?serAddress ?elat ?elong ?sType ?sTypeIta ?sName ?email 

?note ?multimedia ?description ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS  OPTION (inference "urn:ontology"). 

 OPTIONAL {?ser <http://schema.org/name> ?sName} 

 ?ser <http://schema.org/streetAddress> ?serAddress. 

 OPTIONAL {?ser skos:note ?note} 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it") } 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia} 

 OPTIONAL {?ser schema:email ?email } 

 { 

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

#VOS  ?entry geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?entry omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } UNION { 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

#VOS  ?ser geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?ser omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } 

 { ?ser a km4c:Accommodation 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:CulturalActivity 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:TourismService 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:WineAndFood 

#VOS OPTION (inference "urn:ontology"). 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 



 ?sType rdfs:label ?sTypeIta. FILTER(LANG(?sTypeIta)="it") 

} 

2.15 Service-Htl-B&B-latlng 
retrieves all the services in the Hotel and Bed&Breakfast classes within a radius %DIST (km) from a position 

%LAT %LNG 

#ID: service-htl-b&b-latlng-%DISTkm 
PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX schema:<http://schema.org/> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

PREFIX dc:<http://purl.org/dc/elements/1.1/> 

SELECT DISTINCT ?ser ?serAddress ?elat ?elong ?sType ?sTypeIta ?sName ?email 

?note ?multimedia ?description ?x WHERE { 

 ?ser rdf:type km4c:Service 

#VOS  OPTION (inference "urn:ontology"). 

 OPTIONAL {?ser <http://schema.org/name> ?sName} 

 ?ser <http://schema.org/streetAddress> ?serAddress. 

 OPTIONAL {?ser skos:note ?note} 

 OPTIONAL {?ser dc:description ?description FILTER(LANG(?description) = "it") } 

 OPTIONAL {?ser km4c:multimediaResource ?multimedia} 

 OPTIONAL {?ser schema:email ?email } 

 { 

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

#VOS  ?entry geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?entry omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } UNION { 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

#VOS  ?ser geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

#OIM  ?ser omgeo:nearby(%LAT %LNG "%DISTkm") . 

 } 

 { ?ser a km4c:Hotel 

#VOS OPTION (inference "urn:ontology"). 

 } 

UNION { ?ser a km4c:Bed_and_breakfast 

#VOS OPTION (inference "urn:ontology"). 

 } 

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?sTypeIta. FILTER(LANG(?sTypeIta)="it") 

} 



2.16 Service-text-florence 
retrieves all services in the municipality of Florence matching the keyword %WORD. 

#ID:service-text-firenze 

PREFIX luc: <http://www.ontotext.com/owlim/lucene#> 

PREFIX km4c:<http://www.disit.org/km4city/schema#>  

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#>  

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#>  

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#>  

PREFIX schema:<http://schema.org/#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/>  

SELECT DISTINCT ?ser ?elong ?elat ?sTypeIta WHERE { 

 ?ser ?p ?txt.   

#OIM ?txt luc:myIndex "%WORD". 

#OIM ?txt luc:score ?sc. 

#VOS ?txt bif:contains "%WORD" OPTION (score ?sc). 

 {  

  ?ser km4c:hasAccess ?entry.  

  ?entry geo:lat ?elat.  

  ?entry geo:long ?elong.  

  ?nc km4c:hasExternalAccess ?entry.  

  ?nc km4c:belongToRoad ?road.  

  ?road km4c:inMunicipalityOf ?mun.  

  ?mun foaf:name "FIRENZE". 

 } UNION {  

  ?ser km4c:isInRoad ?road. 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

  ?road km4c:inMunicipalityOf ?mun.  

  ?mun foaf:name "FIRENZE". 

 }UNION {  

  ?ser km4c:isInMunicipality ?com.  

  ?com foaf:name "FIRENZE". 

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

 }  

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?sTypeIta. FILTER(LANG(?sTypeIta)="it") 

} 

2.17 Service-text-latlng 
retrieves all services matching a keyword in any field given a position %LAT %LNG and a radius of 500m.  

#ID:service-text-latlng 

PREFIX luc: <http://www.ontotext.com/owlim/lucene#> 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 



PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

SELECT DISTINCT ?ser ?elong ?elat ?sTypeIta WHERE { 

 ?ser ?p ?txt.   

#OIM ?txt luc:myIndex "%WORD". 

#OIM ?txt luc:score ?sc. 

#VOS ?txt bif:contains "%WORD" OPTION (score ?sc). 

 {  

  ?ser km4c:hasAccess ?entry. 

  ?entry geo:lat ?elat. 

  ?entry geo:long ?elong. 

#OIM  ?entry omgeo:nearby(%LAT %LNG "0.5km"). 

#VOS  ?entry geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=0.5) 

 }UNION{  

  ?ser geo:lat ?elat. 

  ?ser geo:long ?elong. 

#OIM  ?ser omgeo:nearby(%LAT %LNG "0.5km"). 

#VOS  ?ser geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=0.5) 

 }  

#VOS ?ser a ?sType. FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

#OIM ?ser <http://www.openrdf.org/schema/sesame#directType> ?sType. 

FILTER(?sType!=km4c:RegularService && ?sType!=km4c:Service) 

 ?sType rdfs:label ?sTypeIta. FILTER(LANG(?sTypeIta)="it") 

} 

2.18 Sensor-florence  
retrieves all the sensors within the municipality of Florence. 

#ID:sensor-florence 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX schema:<http://schema.org/> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

SELECT DISTINCT ?sensor ?idSensore ?lat ?long ?address ?x WHERE { 

 ?sensor rdf:type km4c:SensorSite. 

 ?sensor geo:lat ?lat. 

 ?sensor geo:long ?long. 

 ?sensor dcterms:identifier ?idSensore. 

 ?sensor km4c:placedOnRoad ?road. 

 ?road km4c:inMunicipalityOf ?mun. 

 ?mun foaf:name "FIRENZE". 

 ?sensor schema:streetAddress ?address. 

} 

2.19 Sensor-latlng 
retrieves all sensors within a radius %DIST (km) from a position %LAT %LNG. 

#ID:sensor-latlng-%DISTkm 



PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX dct:<http://purl.org/dc/terms/#> 

PREFIX omgeo:<http://www.ontotext.com/owlim/geo#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX skos:<http://www.w3.org/2004/02/skos/core#> 

PREFIX rdfs:<http://www.w3.org/2000/01/rdf-schema#> 

SELECT DISTINCT ?sensor ?idSensore  ?lat ?long ?address ?x WHERE { 

 ?sensor rdf:type km4c:SensorSite. 

 ?sensor geo:lat ?lat. 

 ?sensor geo:long ?long. 

 ?sensor dcterms:identifier ?idSensore. 

#OIM ?sensor omgeo:nearby(%LAT %LNG "%DISTkm"). 

#VOS ?sensor geo:geometry ?geo. 

#VOS  filter(bif:st_distance (?geo, bif:st_point (%LNG,%LAT))<=%DIST) 

 ?sensor schema:streetAddress ?address. 

} 

2.20 Sensor-status 
retrieves the latest information associated with the sensor %SENSOR. 

 

#ID:sensor-status 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT  ?avgDistance ?avgTime ?occupancy ?concentration ?vehicleFlow 

?averageSpeed ?thresholdPerc ?speedPercentile ?timeInstant WHERE{ 

 ?sensor rdf:type km4c:SensorSite. 

 ?sensor dcterms:identifier "%SENSOR". 

 ?sensor km4c:hasObservation ?obs. 

 ?obs km4c:measuredTime ?time. 

 ?time dcterms:identifier ?timeInstant. 

 OPTIONAL {?obs km4c:averageDistance ?avgDistance} 

 OPTIONAL {?obs km4c:averageTime ?avgTime} 

 OPTIONAL {?obs km4c:occupancy ?occupancy} 

 OPTIONAL {?obs km4c:concentration ?concentration} 

 OPTIONAL {?obs km4c:vehicleFlow ?vehicleFlow} 

 OPTIONAL {?obs km4c:averageSpeed ?averageSpeed} 

 OPTIONAL {?obs km4c:thresholdPerc ?thresholdPerc} 

 OPTIONAL {?obs km4c:speedPrecentile ?speedPercentile} 

} ORDER BY DESC (?timeInstant) LIMIT 1 

2.21 Sensor-distribution 
finds for each day the count of sensor status updates received. 

#ID:sensor-distribution 

PREFIX dcterms:<http://purl.org/dc/terms/> 



PREFIX km4c:<http://www.disit.org/km4city/schema#> 

SELECT ?yy ?mm ?dd (COUNT(*) AS ?c) where { 

#VOS { 

#VOS  SELECT DISTINCT * where { 

 ?x a km4c:Observation. 

 ?x km4c:measuredTime/dcterms:identifier ?d. 

#VOS  } 

#VOS } 

 BIND( year(?d) as ?yy) 

 BIND( month(?d) as ?mm) 

 BIND( day(?d) as ?dd) 

}   

GROUP BY ?yy ?mm ?dd ORDER BY ?yy ?mm ?dd 

2.22 Parking-status 
retrieves the latest information associated with the parking lot %CAR_PARK. 

#ID:parking-status 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX time:<http://www.w3.org/2006/time#> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

SELECT DISTINCT  ?situationRecord ?instantDateTime ?occupancy ?free ?occupied 

?capacity WHERE { 

  ?park rdf:type km4c:Car_park. 

  ?park schema:name "%CAR_PARK". 

  ?cps km4c:observeCarPark ?park. 

  ?cps km4c:capacity ?capacity. 

  ?situationRecord km4c:relatedToSensor ?cps. 

  ?situationRecord km4c:observationTime ?time. 

  ?time dcterms:identifier ?instantDateTime. 

  ?situationRecord km4c:parkOccupancy ?occupancy. 

  ?situationRecord km4c:free ?free. 

  ?situationRecord km4c:occupied ?occupied. 

} ORDER BY DESC (?instantDateTime) LIMIT 1 

2.23 Parking-distribution 
retrieves for each day the count of parking status records acquired. 

#ID:parking-distribution 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

SELECT ?yy ?mm ?dd (COUNT(*) AS ?c) where { 

#VOS { 

#VOS  SELECT DISTINCT * where { 

   ?x a km4c:SituationRecord. 

   ?x km4c:observationTime/dcterms:identifier ?d. 

#VOS  } 

#VOS } 

 BIND( year(?d) as ?yy) 

 BIND( month(?d) as ?mm) 

 BIND( day(?d) as ?dd) 



} 

GROUP BY ?yy ?mm ?dd ORDER BY ?yy ?mm ?dd 

2.24 Weather-florence 
retrieves the latest forecast available for the municipality of Florence. 

#ID:weather-florence 

PREFIX geo:<http://www.w3.org/2003/01/geo/wgs84_pos#> 

PREFIX foaf:<http://xmlns.com/foaf/0.1/> 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

PREFIX xsd:<http://www.w3.org/2001/XMLSchema#> 

PREFIX rdf:<http://www.w3.org/1999/02/22-rdf-syntax-ns#> 

PREFIX schema:<http://schema.org/> 

PREFIX time:<http://www.w3.org/2006/time#> 

SELECT ?giorno ?descrizione ?minTemp ?maxTemp ?instantDateTime ?wPred WHERE { 

 { 

  SELECT DISTINCT ?wRep ?instantDateTime WHERE { 

   ?comune rdf:type km4c:Municipality. 

   ?comune foaf:name "FIRENZE". 

   ?comune km4c:hasWeatherReport ?wRep. 

   ?wRep km4c:updateTime ?instant. 

   ?instant schema:value ?instantDateTime. 

  } ORDER BY DESC (?instantDateTime) LIMIT 1  

 } 

 ?wRep km4c:hasPrediction ?wPred. 

 ?wPred dcterms:description ?descrizione. 

 ?wPred km4c:day  ?giorno. 

 ?wPred km4c:hour "giorno"^^xsd:string. 

 OPTIONAL { ?wPred km4c:minTemp ?minTemp.} 

 OPTIONAL { ?wPred km4c:maxTemp ?maxTemp.} 

} 

2.25 Weather-distribution 
retrieves for each day the count of weather forecasts acquired. 

#ID:meteo-distribution 

PREFIX dcterms:<http://purl.org/dc/terms/> 

PREFIX km4c:<http://www.disit.org/km4city/schema#> 

SELECT ?yy ?mm ?dd (COUNT(*) AS ?c) where { 

 ?x a km4c:WeatherReport. 

 ?x km4c:updateTime/<http://schema.org/value> ?d. 

 BIND( year(?d) as ?yy) 

 BIND( month(?d) as ?mm) 

 BIND( day(?d) as ?dd) 

}   

GROUP BY ?yy ?mm ?dd ORDER BY ?yy ?mm ?dd 

3 Parameters data values 
In the following sections are reported the values to be used in the queries 

3.1 %LAT %LNG 
43.778159022442345;11.255836486816406 



43.774812450590474;11.256265640258789 

43.77326304834337;11.24751091003418 

43.770721941755674;11.261930465698242 

3.2 %LINE 
10 

2 

1 

11 

12 

13 

14 

15 

17 

19 

20 

21 

22 

23 

24 

25 

26 

27 

28 

29 

3 

30 

303 

31 

35 

36 

38 

4 

40 

41 

42 

43 

44 

45 

46 

47 

48 

49 

5 

50 

52 

54 

56 

57 

59 

6 

60 

61 

62 

63 

64 

66 

7 

72 

73 

74 

75 

76 

77 

78 

8 

81 

82 

83 

84 

85 

86 

89 

9 

90 

91 

92 

93 

94 

C1 

C2 

C3 

D 

G 

M 

R 

S1 

S3 

SC 

SF 

3.3 %BUS_STOP 
SAN ZANOBI 

SANTA REPARATA 

SCALA 

STAZIONE PENSILINA 

PIAZZA DI SAN MARCO 

VENEZIA 

VIA DELLE CARRA 

MILLE 07 

CAIROLI 

MILLE 02 

GIACOMINI 

PACINOTTI 

INDIPENDENZA 

FRA' BARTOLOMMEO 01 

SAVONAROLA 

CONTI 

VASARI 

FICINO 

PICO DELLA MIRANDOLA 

FRA' BARTOLOMMEO 

LA MARMORA 

BRONZETTI 

NICOLODI 

BALDESI 

CENTO STELLE 

PIERLUIGI DA PALESTRINA 

BENEDETTO MARCELLO 

TOSELLI BELLINI 

VIALE VERGA 

DUSE 04 

DUSE 02 

SAFFI 

FRATELLI ROSSELLI 

TARTINI 01 

PERGOLESI 01 

MANIFATTURA TABACCHI 

CASCINE 

LE CASCINE 

BOITO 

PERGOLESI 

TARTINI 04 

TARTINI 

KENNEDY 

CASCINE 02 

PAISIELLO 

SETTE SANTI 

CARNESECCHI 

SAN GERVASIO 

DUSE 

DUSE 03 

VERGA 01 

VERGA 03 

PALAZZESCHI 

RIDOLFI 

ARAZZIERI 

PIAZZA PUCCINI 

TOSELLI 

PASSERELLA 

RINUCCINI 

SQUARCIALUPI 



VIVALDI 

STAZIONE SCALETTE 

ROSSELLI 

STAZIONE VALFONDA 

SANTI FIORENTINI 

DON GIOVANNI VERITA' 

STAZIONE ABSIDE S.M.N. 

STAZIONE PIAZZA DELL'UNITA' 

D'ANNUNZIO 04 

FILAROCCA 

DE AMICIS 

D'ANNUNZIO 06 

FANTI 

CALZA 02 

ROMANA 

PIAZZA SAN FELICE 

COMPARETTI 

PASQUALI 

MANNI 06 

MANNI 04 

MANNI 02 

CAMPO D'ARRIGO 04 

CAMPO D'ARRIGO 02 

PODESTA' 

SENESE 

SENESE 14 

I ROSSI 02 

GLI ALLORI 

DUE STRADE 02 

SAN GAGGIO 

SAN GAGGIO 04 

SAN GAGGIO 02 

VILLA TORRE ROSSA 

LAVAGNINI LIBERTA' 

VIGNA NUOVA 

VECCHIETTI 

FRESCOBALDI 

GUICCIARDINI 

D?ANNUNZIO RONDINELLA 

RONDINELLA 03 

NOVELLI 

OSPEDALE TORRE GALLI 

DI SCANDICCI 04 

DI SCANDICCI 02 

FILARETE SCUOLE 

FILARETE 02 

MASOLINO 02 

PISANA 08 

PISANA 06 

VANNI 

ROSSO FIORENTINO 

CINEMA UNIVERSALE 

PERGOLA 

MASOLINO 

STARNINA 01 

DI SCANDICCI 

DI SCANDICCI 03 

DI SCANDICCI 05 

DI SCANDICCI 07 

DI SCANDICCI TORRE GALLI 

LUNGO L'AFFRICO 

RAMAZZINI 

RONDINELLA 

ARCIPRESSI 

ANTONIO DEL POLLAIOLO 05 

SCIPIONE AMMIRATO 

LANDUCCI 

BOVIO 

SEGNI 

MATTONAIA 

D'AZEGLIO 

COLONNA 

MUSEO ARCHEOLOGICO 

MUSEO DI SAN MARCO 

PESCAIA DI SANTA ROSA 

SODERINI 

FONDERIA 

SANTA MARIA AL PIGNONE 

AMMANNATI 

VIA ANTONIO DEL POLLAIOLO 

PIER DELLA FRANCESCA 03 

ALBERTINELLI 

VIA PISANA 

PISANA 10 

MATTEO DI GIOVANNI 

CESTELLO 

SAN FREDIANO 

SANTISSIMA ANNUNZIATA 

COLONNA 01 

NICCOLINI 

LEOPARDI 

MAZZINI 01 

OBERDAN 

FRA' PAOLO SARPI 

VILLA ARRIVABENE 

PIER DELLA FRANCESCA 

FRANCAVILLA 

OLIVUZZO 

PIAZZA SANTA MARIA NOVELLA 

NENNI 02 

DI SCANDICCI DON GNOCCHI 

OSPEDALE DON GNOCCHI 

GIOVANNI DELLA CASA 

T1 ALAMANNI 

CURTATONE 

RUCELLAI 

VITTORIO VENETO 

VIA GRAMIGNANO 



PIAZZA TASSO 

ALEARDI 

ARNOLFO 

ORCAGNA 

STAGNI DI FOCOGNANO 

BUOZZI 

SANTA MARIA 

LA VILLA 

OLMO 

CONFINI 

TOSCA FIESOLI 

TOSCA FIESOLI 02 

PALAGETTA 04 

PARADISO 

CARMINE 

ROSSINI 

CAMPI BISENZIO TINTORI 

DON GNOCCHI 

GIORDANO BRUNO 

BUONARROTI 

PIAZZA GRAMSCI 

STAZIONE GALLERIA 

SANT'AGOSTINO 

LEOPOLDA 

IL PRATO 

FUSINATO 

IL CIONFO 

SAN MARCO VECCHIO 

PALANCOLA 

PONTE ALLE RIFFE 

SAN MARCO VECCHIO FS 

BORGHINI 

MORGAGNI 05 

MORGAGNI 07 

OSPEDALI DI CAREGGI 

LEONE X 

MORGAGNI 06 

MARITI 

MULINO BIONDI 

BRAMBILLA 

PORTA SAN FREDIANO 

VOLTA 01 

FERRARIS 

PIAZZA EDISON 

DI SAN DOMENICO 

FRULLINO 

FORESE DONATI 

DI SAN DOMENICO 07 

DI SAN DOMENICO 09 

OSPEDALE CAMERATA 01 

SAN DOMENICO 01 

MANTELLINI 

MANTELLINI 03 

FIESOLE PIAZZA MINO 

STAZIONE LARGO ALINARI 

BALDINUCCI 

BONAINI 

MILANESI 

PALAZZO BRUCIATO 

PISACANE 

DALMAZIA 01 

GORE 

PASSAVANTI 

FAENTINA 36 

FAENTINA 34 

FAENTINA 32 

FAENTINA 30 

PIAN DI MUGNONE 02 

FAENTINA 26 

FAENTINA 24 

CALDERAIO 02 

FAENTINA 20 

MANZOLO 02 

PONTE ALLA BADIA 02 

PONTE DI SAN DONATO 

TERZOLLE 

GALOPPO 

PIOVANI 

CORSALI 

FONTE DI NARCISO 

LA PIRA 

CASTAGNETO 

SALVIATI FS 

MORGAGNI 04 

DALMAZIA 02 

STATUTO 

CADORNA STATUTO 

GIOIA 

GIORGINI 

MONTELATICI 

CELSO 

MERCATI 

CAPPUCCINI 

MASSAIA 

MASSAIA 02 

VITTORIO EMANUELE 

BIGOZZI 

STATUTO 04 

GUIDO MONACO 

PAGNINI FABBRONI 

PAGNINI STATUTO FS 

LAMI STATUTO FS 

STAZIONE MERCATO CENTRALE 

LIBERTA' 

LUCA GIORDANO 

DEPOSITO CURE 

VIALE DEI MILLE 

MILLE



3.4 %CAR_PARK 
G. Boccaccio 

B. Buozzi 

A. Gramsci 

G. Guerra 

G. Leopardi 

Bisarnella 

Garage La Stazione Spa 

Piazza del Mercato Centrale 

Parcheggio Sant'Ambrogio 

Parcheggio Piazza Beccaria 

Parcheggio Parterre 

Fortezza Fiera 

Piazzale della Porta al Prato 

Parcheggio Oltrarno 

Parcheggio Viale Gaetano Pieraccini 

Parcheggio Viale Europa 

Piazzale Montelungo 

3.5 %SENSOR 
EM0100401 

EM0100402 

EM0100501 

EM0100502 

EM0100601 

EM0100602 

EM0100801 

EM0100802 

FI055SPI00502 

FI055SPI00501 

FI055ZTL00801 

EM0100201 

EM0100202 

EM0100301 

EM0100302 

EM0100101 

EM0100102 

3.6 %WORD 
casa 

marte 

michelangelo 

leonardo 

pitti 


